

COMPSCI 111 / 111G

Mastering Cyberspace:
An introduction to practical computing

Spreadsheets

Revision

Ticket Sales

- Check if more than 90% of the tickets were sold, or if less than 50% of the tickets were sold. In either case, a new venue is required next time.
- What formula in cell G9?

	A	B	C	D	E	F	G	H
1	Tickets sales							
2								
3	Price	\$10.00						
4								
5	Event	Tickets Available	Tickets Sold	Remaining	Sales		Different venue required	
6	Cycling	4000	2000	2000	\$20,000.00		No	
7	Weightlifting	2000	750	1250	\$7,500.00		Yes	
8	Triathlon	1000	100	900	\$1,000.00		Yes	
9	Soccer	3000	3000	0	\$30,000.00		Yes	
10	Badminton	5000	4500	500	\$45,000.00		No	
11		15000	10350	4650	\$103,500.00			

Exercise

Examine the following spreadsheet that keeps track of beetle races. Each beetle is involved in three races, and the time it took to cross the finish line is recorded. The best time out of the three races is calculated and will be used to determine the overall winner.

Some races are aborted before the beetle finishes, so no time is recorded. Occasionally, a beetle will escape from the track, and is therefore disqualified from the races (recorded as a "D").

Write down the formula used in cell E4

	A	B	C	D	E
1	Beetle racing				
2		Race			
3	Beetle	1	2	3	Best
4	George	12		46	12
5	John		43	35	35
6	Paul		32	33	32
7	Ringo	19	28	D	Disqualified
8	Juice	23	13		13
9	VW	34	D		Disqualified

Looking up values in a table

Often have tables of data

- We want to look up a value
- e.g. given ID number, what is the name?

Student ID	Name	Phone
9100983	Andrew	123-4567
2098382	Albert	234-7654
2289483	Adrienne	321-7839
2109374	Ann	567-8932

Use a lookup formula

- VLOOKUP - looking up values in a vertical table
- HLOOKUP - looking up values in a horizontal table

VLOOKUP

VLOOKUP(value, table, column, [range])

Value.

This is the value we already have written down. We want to use this value to look up a corresponding value in a table.

Range of cells.

This is the table we are using to look up the value in.

Usually we want to use absolute references for the table.

Number.

This specifies which column in the table contains the data we want.

Boolean value.

True if we want to match a range of values

False if we want an exact match.

Example

	A	B	C	D	E	F	G	H
1								
2		Students Enrolled			ID	UPI	Name	
3		ID	Name		199444	jhub001	Jacob	
4		800526	Ethan		303114	mkop032	Michael	
5		952348	William		465336	jjs012	Joshua	
6		303114	Michael		769666	mwen003	Matthew	
7		973748			800526	eupt008	Ethan	
8					812069	acut017	Andrew	
9					887268	dden011	Daniel	
10					952348	whur034	William	
11					973748	jfr002	Joseph	
12					997073	cca005	Christopher	
13								

=VLOOKUP(value, table, column, range)

	A	B	C	D	E	F	G	H
1								
2		Students Enrolled			ID	UPI	Name	
3		ID	Name		199444	jhub001	Jacob	
4		800526	Ethan		303114	mkop032	Michael	
5		952348	William		465336	jjs012	Joshua	
6		303114	Michael		769666	mwen003	Matthew	
7		973748			800526	eupt008	Ethan	
8					812069	acut017	Andrew	
9					887268	dden011	Daniel	
10					952348	whur034	William	
11					973748	jfr002	Joseph	
12					997073	cca005	Christopher	
13								

False

1 2 3

Example

Use a VLOOKUP to find the description for a recorded wind speed

	A	B	C	D	E	F	G
23					Beaufort Scale		
24					Speed (km/hr)	Beaufort number	Description
25	Day	Wind Spd	Description		0	0	Calm
26	Mon	20	Moderate breeze		1	1	Light air
27	Tues	5	Light air		2	2	Light breeze
28	Wed	0	Calm		3	3	Gentle breeze
29	Thurs	15	Gentle breeze		4	4	Moderate breeze
30	Fri	20	Moderate breeze		5	5	Fresh breeze
31	Sat	40	Strong breeze		6	6	Strong breeze
32	Sun	78	Strong gale		7	7	Near gale
33					8	8	Gale
34					9	9	Strong gale
35					10	10	Storm
36					11	11	Violent storm
37					12	12	Hurricane

=VLOOKUP(value, table, column, range)

=VLOOKUP(B26, \$E\$25:\$G\$37, 3, TRUE)

HLOOKUP

Same as VLOOKUP, but for horizontal tables

HLOOKUP(value, table, row, [range])

Value.

This is the value we already have written down. We want to use this value to look up a corresponding value in a table.

Range of cells.

This is the table we are using to look up the value in.

Usually we want to use absolute references for the table.

Number.

This specifies which row in the table contains the data we want.

Boolean value.

True if we want to match a range of values

False if we want an exact match.

Exercise

What formula would be used in cell E10?

- Use an IF and an HLOOKUP

	A	B	C	D	E	F	G	H	I
1									
2		Movie Prices							
3			Mon	Tues	Wed	Thurs	Fri	Sat	Sun
4		Before 5pm	\$14.00	\$8.50	\$14.00	\$15.00	\$15.00	\$15.00	\$15.00
5		After 5pm	\$14.00	\$8.50	\$14.00	\$15.00	\$15.00	\$15.00	\$15.00
6									
7									
8		Movie Tickets							
9		<i>Name</i>	<i>Day</i>	<i>Evening</i>	<i>Cost</i>				
10		Helen	Fri	Y	\$15.00				
11		Michael	Tues	N	\$8.50				
12		Steve	Mon	N	\$14.00				
13		Ruth	Mon	N	\$14.00				
14		David	Sat	Y	\$15.00				
15		Nania	Sat	Y	\$15.00				
16									
17				Total	\$81.50				

Exercise: ThinkGeek T-Shirts



<http://www.thinkgeek.com/>

Exercise

What formulae should be used in cells D15, E15, F15 and F26?

	A	B	C	D	E	F
1	T-Shirt Sizes				T-Shirt Prints	
2	<i>Size</i>	<i>Price</i>			<i>Code</i>	<i>Description</i>
3	S	\$ 10.99			1001	2 + 2 = 5
4	M	\$ 11.99			1010	geek inside
5	L	\$ 12.99			1011	<BODY>
6	XL	\$ 13.99			1100	man woman
7	XXL	\$ 14.99			1101	obey gravity
8	XXXL	\$ 15.99			1110	I'm blogging this
9					1111	Arrrrrgh...
10						
11						
12						
13	Invoice					
14	<i>Code</i>	<i>Size</i>	<i>Number</i>	<i>Description</i>	<i>Price</i>	<i>Cost</i>
15	1010	M	1	geek inside	\$ 11.99	\$ 11.99
16	1010	L	1	geek inside	\$ 12.99	\$ 12.99
17	1011	S	3	<BODY>	\$ 10.99	\$ 32.97
18	1110	XL	1	I'm blogging this	\$ 13.99	\$ 13.99
19	1001	XL	1	2 + 2 = 5	\$ 13.99	\$ 13.99
20	1101	M	2	obey gravity	\$ 11.99	\$ 23.98
21	1111	M	1	Arrrrrgh...	\$ 11.99	\$ 11.99
22						
23						
24						
25						
26					Total	\$ 121.90

A Complicated Example

	A	B	C	D	E	F	G	H	I	J	K	L
1												
2		Day Pricing (Before 5pm)										
3			Mon	Tues	Wed	Thurs	Fri	Sat	Sun		<i>Code</i>	<i>Description</i>
4		Adult	\$11.00	\$8.50	\$11.00	\$14.00	\$14.00	\$15.00	\$15.00		A	Adult
5		Child	\$8.50	\$7.00	\$8.50	\$8.50	\$8.50	\$9.00	\$9.00		C	Child
6		Secondary Student	\$11.00	\$8.50	\$11.00	\$12.00	\$12.00	\$13.00	\$13.00		SS	Secondary Student
7		Tertiary Student	\$11.00	\$8.50	\$11.00	\$12.00	\$12.00	\$15.00	\$15.00		TS	Tertiary Student
8		Senior Citizen	\$8.50	\$7.00	\$8.50	\$8.50	\$8.50	\$9.00	\$9.00		SC	Senior Citizen
9		Beneficiaries	\$8.50	\$7.00	\$8.50	\$8.50	\$8.50	\$15.00	\$15.00		B	Beneficiaries
10												
11		Evening Pricing (After 5pm)										
12			Mon	Tues	Wed	Thurs	Fri	Sat	Sun			
13		Adult	\$14.00	\$8.50	\$14.00	\$15.00	\$15.00	\$15.00	\$15.00			
14		Child	\$8.50	\$7.00	\$8.50	\$9.00	\$9.00	\$9.00	\$9.00			
15		Secondary Student	\$12.00	\$8.50	\$12.00	\$13.00	\$13.00	\$13.00	\$13.00			
16		Tertiary Student	\$12.00	\$8.50	\$12.00	\$13.00	\$13.00	\$15.00	\$15.00			
17		Senior Citizen	\$8.50	\$7.00	\$8.50	\$9.00	\$9.00	\$9.00	\$9.00			
18		Beneficiaries	\$14.00	\$7.00	\$14.00	\$15.00	\$15.00	\$15.00	\$15.00			
19												
20		Movie Tickets										
21		<i>Name</i>	<i>Code</i>	<i>Day</i>	<i>Evening</i>	<i>Cost</i>						
22		Helen	A	Fri	Y	\$15.00						
23		Michael	TS	Tues	N	\$8.50						
24		Steve	SS	Mon	N	\$11.00						
25		Ruth	SC	Mon	N	\$8.50						
26		David	B	Sat	Y	\$15.00						
27		Nania	C	Sat	Y	\$9.00						
28												
29					Total	\$67.00						

Calculate the cost of a movie ticket



Example (1)

Start by assuming that it is easy

- Assume that the tickets are all morning tickets (ignore time of day)
- Assume that all tickets are adult tickets (ignore different prices)
- Assume that all tickets are for Monday movies (ignore day)

	A	B	C	D	E	F	G	H	I
1									
2		Day Pricing (Before 5pm)							
3		Mon	Tues	Wed	Thurs	Fri	Sat	Sun	
4		Adult	\$11.00	\$8.50	\$11.00	\$14.00	\$14.00	\$15.00	\$15.00
5		Child	\$8.50	\$7.00	\$8.50	\$8.50	\$9.00	\$9.00	
6		Secondary Student	\$11.00	\$8.50	\$11.00	\$12.00	\$12.00	\$13.00	\$13.00
7		Tertiary Student	\$11.00	\$8.50	\$11.00	\$12.00	\$12.00	\$15.00	\$15.00
8		Senior Citizen	\$8.50	\$7.00	\$8.50	\$8.50	\$9.00	\$9.00	
9		Beneficiaries	\$8.50	\$7.00	\$8.50	\$8.50	\$8.50	\$15.00	\$15.00

	A	B	C	D	E	F
20		Movie Tickets				
21		Name	Code	Day	Evening	Cost
22		Helen	A	Fri	Y	\$11.00
23		Michael	TS	Tues	N	
24		Steve	SS	Mon	N	
25		Ruth	SC	Mon	N	
26		David	B	Sat	Y	
27		Nania	C	Sat	Y	

=C4

Example (2)

Look up the category (Adult, Tertiary Student, etc.)

	J	K	L
1			
2		Code	Description
3		A	Adult
4		C	Child
5		SS	Secondary Student
6		TS	Tertiary Student
7		SC	Senior Citizen
8		B	Beneficiaries

	B	C	D	E	F
20		Movie Tickets			
21		Name	Code	Day	Evening
22		Helen	A	Fri	Y
23		Michael	TS	Tues	N
24		Steve	SS	Mon	N
25		Ruth	SC	Mon	N
26		David	B	Sat	Y
27		Nania	C	Sat	Y

=VLOOKUP(C22, \$K\$3:\$L\$8, 2, FALSE)

Example (3)

Use that category to look up the cost of the ticket

	B	C	D	E	F	G	H	I	
2		Day Pricing (Before 5pm)							
3		Mon	Tues	Wed	Thurs	Fri	Sat	Sun	
4		Adult	\$11.00	\$8.50	\$11.00	\$14.00	\$14.00	\$15.00	\$15.00
5		Child	\$8.50	\$7.00	\$8.50	\$8.50	\$9.00	\$9.00	
6		Secondary Student	\$11.00	\$8.50	\$11.00	\$12.00	\$12.00	\$13.00	\$13.00
7		Tertiary Student	\$11.00	\$8.50	\$11.00	\$12.00	\$12.00	\$15.00	\$15.00
8		Senior Citizen	\$8.50	\$7.00	\$8.50	\$8.50	\$9.00	\$9.00	
9		Beneficiaries	\$8.50	\$7.00	\$8.50	\$8.50	\$8.50	\$15.00	\$15.00

	B	C	D	E	F
20		Movie Tickets			
21		Name	Code	Day	Evening
22		Helen	A	Fri	Y
23		Michael	TS	Tues	N
24		Steve	SS	Mon	N
25		Ruth	SC	Mon	N
26		David	B	Sat	Y
27		Nania	C	Sat	Y

=VLOOKUP(VLOOKUP(C22, \$K\$3:\$L\$8, 2, FALSE),
\$B\$4:\$I\$9, 2, FALSE)

Example (4)

We need to figure out how to find the right column

- Column depends on the day
- VLOOKUP function needs to know the column number
- So we need some method of looking up a value and finding the column

Use the Help

- Lookup functions
- MATCH

MATCH(lookup_value,lookup_array,match_type)

Returns the relative position of an item in an array that matches a specified value in a specified order.

Example (5)

Test the function

	B	C	D	E	F	G	H	I
2	Day Pricing (Before 5pm)							
3		Mon	Tues	Wed	Thurs	Fri	Sat	Sun
4	Adult	\$11.00	\$8.50	\$11.00	\$14.00	\$14.00	\$15.00	\$15.00
5	Child	\$8.50	\$7.00	\$8.50	\$8.50	\$8.50	\$9.00	\$9.00
6	Secondary Student	\$11.00	\$8.50	\$11.00	\$12.00	\$12.00	\$13.00	\$13.00
7	Tertiary Student	\$11.00	\$8.50	\$11.00	\$12.00	\$12.00	\$15.00	\$15.00
8	Senior Citizen	\$8.50	\$7.00	\$8.50	\$8.50	\$8.50	\$9.00	\$9.00
9	Beneficiaries	\$8.50	\$7.00	\$8.50	\$8.50	\$8.50	\$15.00	\$15.00

	B	C	D	E	F	G	H
20	Movie Tickets						
21	Name	Code	Day	Evening	Cost		Testing
22	Helen	A	Fri	Y	Adult		5
23	Michael	TS	Tues	N			2
24	Steve	SS	Mon	N			1
25	Ruth	SC	Mon	N			1
26	David	B	Sat	Y			6
27	Nania	C	Sat	Y			6

=MATCH(D22, \$C\$3:\$I\$3, 0)

Example (6)

Use the MATCH function to get the correct column

	B	C	D	E	F	G	H	I
2	Day Pricing (Before 5pm)							
3		Mon	Tues	Wed	Thurs	Fri	Sat	Sun
4	Adult	\$11.00	\$8.50	\$11.00	\$14.00	\$14.00	\$15.00	\$15.00
5	Child	\$8.50	\$7.00	\$8.50	\$8.50	\$8.50	\$9.00	\$9.00
6	Secondary Student	\$11.00	\$8.50	\$11.00	\$12.00	\$12.00	\$13.00	\$13.00
7	Tertiary Student	\$11.00	\$8.50	\$11.00	\$12.00	\$12.00	\$15.00	\$15.00
8	Senior Citizen	\$8.50	\$7.00	\$8.50	\$8.50	\$8.50	\$9.00	\$9.00
9	Beneficiaries	\$8.50	\$7.00	\$8.50	\$8.50	\$8.50	\$15.00	\$15.00

	B	C	D	E	F
20	Movie Tickets				
21	Name	Code	Day	Evening	Cost
22	Helen	A	Fri	Y	\$14.00
23	Michael	TS	Tues	N	\$8.50
24	Steve	SS	Mon	N	\$11.00
25	Ruth	SC	Mon	N	\$8.50
26	David	B	Sat	Y	\$15.00
27	Nania	C	Sat	Y	\$9.00

=VLOOKUP(VLOOKUP(C22, \$K\$3:\$L\$8, 2, FALSE),
\$B\$4:\$I\$9,
MATCH(D22, \$C\$3:\$I\$3, 0) + 1,
FALSE)

Example (7)

Evening tickets cost a different amount

- Two different tables to calculate the costs

Decision

- If the ticket is an evening ticket, use evening table, otherwise use the morning table

Plan to use an IF

- =IF(E22="Y", use evening, use morning)

	B	C	D	E	F
20	Movie Tickets				
21	Name	Code	Day	Evening	Cost
22	Helen	A	Fri	Y	\$14.00
23	Michael	TS	Tues	N	\$8.50
24	Steve	SS	Mon	N	\$11.00
25	Ruth	SC	Mon	N	\$8.50
26	David	B	Sat	Y	\$15.00
27	Nania	C	Sat	Y	\$9.00

Example (8)

=IF(E22="Y", use evening, use morning)

Formula for calculating the cost using the morning table is:

=VLOOKUP(VLOOKUP(C22, \$K\$3:\$L\$8, 2, FALSE),
\$B\$4:\$I\$9,
MATCH(D22, \$C\$3:\$I\$3, 0) + 1,
FALSE)

So we use that formula in the IF function

=IF(E22="Y",
use the evening table
,
VLOOKUP(VLOOKUP(C22, \$K\$3:\$L\$8, 2, FALSE),
\$B\$4:\$I\$9,
MATCH(D22, \$C\$3:\$I\$3, 0) + 1,
FALSE)
)

Example (9)

The table for evening pricing is similar to morning

- Formula for calculating the cost using the evening table is similar
- Just need to change the table that we refer to

	B	C	D	E	F	G	H	I
11	<i>Evening Pricing (After 5pm)</i>							
12		Mon	Tues	Wed	Thurs	Fri	Sat	Sun
13	Adult	\$14.00	\$8.50	\$14.00	\$15.00	\$15.00	\$15.00	\$15.00
14	Child	\$8.50	\$7.00	\$8.50	\$9.00	\$9.00	\$9.00	\$9.00
15	Secondary Student	\$12.00	\$8.50	\$12.00	\$13.00	\$13.00	\$13.00	\$13.00
16	Tertiary Student	\$12.00	\$8.50	\$12.00	\$13.00	\$13.00	\$15.00	\$15.00
17	Senior Citizen	\$8.50	\$7.00	\$8.50	\$9.00	\$9.00	\$9.00	\$9.00
18	Beneficiaries	\$14.00	\$7.00	\$14.00	\$15.00	\$15.00	\$15.00	\$15.00

=VLOOKUP(VLOOKUP(C22, \$K\$3:\$L\$8, 2, FALSE),
 \$B\$13:\$I\$18,
 MATCH(D22, \$C\$12:\$I\$12, 0) + 1,
 FALSE)

Example (10)

Final formula

=IF(E22="Y",
 VLOOKUP(VLOOKUP(C22, \$K\$3:\$L\$8, 2, FALSE),
 \$B\$13:\$I\$18,
 MATCH(D22, \$C\$12:\$I\$12, 0) + 1,
 FALSE)
 ,
 VLOOKUP(VLOOKUP(C22, \$K\$3:\$L\$8, 2, FALSE),
 \$B\$4:\$I\$9,
 MATCH(D22, \$C\$3:\$I\$3, 0) + 1,
 FALSE)
)